

A Formalization of Declassification with WHAT&WHERE-Security

Sylvia Grawe, Alexander Lux, Heiko Mantel, Jens Sauer

September 13, 2023

Abstract

Research in information-flow security aims at developing methods to identify undesired information leaks within programs from private sources to public sinks. Noninterference captures this intuition by requiring that no information whatsoever flows from private sources to public sinks. However, in practice this definition is often too strict: Depending on the intuitive desired security policy, the controlled declassification of certain private information (WHAT) at certain points in the program (WHERE) might not result in an undesired information leak.

We present an Isabelle/HOL formalization of such a security property for controlled declassification, namely WHAT&WHERE-security from [2]. The formalization includes compositionality proofs for and a soundness proof for a security type system that checks for programs in a simple while language with dynamic thread creation.

Our formalization of the security type system is abstract in the language for expressions and in the semantic side conditions for expressions. It can easily be instantiated with different syntactic approximations for these side conditions. The soundness proof of such an instantiation boils down to showing that these syntactic approximations imply the semantic side conditions.

This Isabelle/HOL formalization uses theories from the entry Strong-Security (see proof document for details).

Contents

1	Preliminary definitions	2
1.1	Type synonyms	2
2	WHAT&WHERE-security	4
2.1	Definition of WHAT&WHERE-security	4
2.2	Proof technique for compositionality results	7
2.3	Proof of parallel compositionality	8

3	Example language and compositionality proofs	9
3.1	Example language with dynamic thread creation	9
3.2	Proofs of atomic compositionality results	13
3.3	Proofs of non-atomic compositionality results	14
4	Security type system	15
4.1	Abstract security type system with soundness proof	15
4.2	Example language for Boolean and arithmetic expressions . .	17
4.3	Example interpretation of abstract security type system . . .	18

1 Preliminary definitions

1.1 Type synonyms

The formalization is parametric in different aspects. Notably, it is parametric in the security lattice it supports.

For better readability, we use the following type synonyms in our formalization (from the entry Strong-Security):

```
theory Types
imports Main
begin

— type parameters:
— 'exp: expressions (arithmetic, boolean...)
— 'val: values
— 'id: identifier names
— 'com: commands
— 'd: domains
```

This is a collection of type synonyms. Note that not all of these type synonyms are used within Strong-Security - some are used in WHATandWHERE-Security.

type-synonym ('id, 'val) State = 'id \Rightarrow 'val

— type for evaluation functions mapping expressions to a values depending on a state

type-synonym ('exp, 'id, 'val) Evalfunction =
 $'exp \Rightarrow ('id, 'val)$ State \Rightarrow 'val

— define configurations with threads as pair of commands and states
type-synonym ('id, 'val, 'com) TConfig = 'com \times ('id, 'val) State

— define configurations with thread pools as pair of command lists (thread pool) and states

type-synonym ('id, 'val, 'com) TPConfig =

$('com\ list) \times ('id,\ 'val)\ State$

— type for program states (including the set of commands and a symbol for terminating - None)

type-synonym $'com\ ProgramState = 'com\ option$

— type for configurations with program states

type-synonym $('id,\ 'val,\ 'com)\ PSConfig =$
 $'com\ ProgramState \times ('id,\ 'val)\ State$

— type for labels with a list of spawned threads

type-synonym $'com\ Label = 'com\ list$

— type for step relations from single commands to a program state, with a label

type-synonym $('exp,\ 'id,\ 'val,\ 'com)\ TLSteps =$
 $(('id,\ 'val,\ 'com)\ TConfig \times 'com\ Label$
 $\times ('id,\ 'val,\ 'com)\ PSConfig)\ set$

— curried version of previously defined type

type-synonym $('exp,\ 'id,\ 'val,\ 'com)\ TLSteps-curry =$
 $'com \Rightarrow ('id,\ 'val)\ State \Rightarrow 'com\ Label \Rightarrow 'com\ ProgramState$
 $\Rightarrow ('id,\ 'val)\ State \Rightarrow bool$

— type for step relations from thread pools to thread pools

type-synonym $('exp,\ 'id,\ 'val,\ 'com)\ TPSteps =$
 $(('id,\ 'val,\ 'com)\ TPConfig \times ('id,\ 'val,\ 'com)\ TPConfig)\ set$

— curried version of previously defined type

type-synonym $('exp,\ 'id,\ 'val,\ 'com)\ TPSteps-curry =$
 $'com\ list \Rightarrow ('id,\ 'val)\ State \Rightarrow 'com\ list \Rightarrow ('id,\ 'val)\ State \Rightarrow bool$

— define type of step relations for single threads to thread pools

type-synonym $('exp,\ 'id,\ 'val,\ 'com)\ TSteps =$
 $(('id,\ 'val,\ 'com)\ TConfig \times ('id,\ 'val,\ 'com)\ TPConfig)\ set$

— define the same type as TSteps, but in a curried version (allowing syntax abbreviations)

type-synonym $('exp,\ 'id,\ 'val,\ 'com)\ TSteps-curry =$
 $'com \Rightarrow ('id,\ 'val)\ State \Rightarrow 'com\ list \Rightarrow ('id,\ 'val)\ State \Rightarrow bool$

— type for simple domain assignments; 'd has to be an instance of order (partial order)

type-synonym $('id,\ 'd)\ DomainAssignment = 'id \Rightarrow 'd::order$

type-synonym $'com\ Bisimulation-type = (('com\ list) \times ('com\ list))\ set$

— type for escape hatches

type-synonym $('d,\ 'exp)\ Hatch = 'd \times 'exp$

```

— type for sets of escape hatches
type-synonym ('d, 'exp) Hatches = (('d, 'exp) Hatch) set

— type for local escape hatches
type-synonym ('d, 'exp) lHatch = 'd × 'exp × nat

— type for sets of local escape hatches
type-synonym ('d, 'exp) lHatches = (('d, 'exp) lHatch) set

end

```

2 WHAT&WHERE-security

2.1 Definition of WHAT&WHERE-security

The definition of WHAT&WHERE-security is parametric in a security lattice ($'d$) and in a programming language ($'com$).

```

theory WHATWHERE-Security
imports Strong-Security.Types
begin

locale WHATWHERE =
  fixes SR :: ('exp, 'id, 'val, 'com) TLSteps
  and E :: ('exp, 'id, 'val) Evalfunction
  and pp :: 'com ⇒ nat
  and DA :: ('id, 'd::order) DomainAssignment
  and LH :: ('d::order, 'exp) lHatches

begin

— define when two states are indistinguishable for an observer on domain d
definition d-equal :: 'd::order ⇒ ('id, 'val) State
  ⇒ ('id, 'val) State ⇒ bool
where
d-equal d m m' ≡ ∀ x. ((DA x) ≤ d → (m x) = (m' x))

abbreviation d-equal' :: ('id, 'val) State
  ⇒ 'd::order ⇒ ('id, 'val) State ⇒ bool
  ( ( - = - ) )
where
m =d m' ≡ d-equal d m m'

— transitivity of d-equality
lemma d-equal-trans:
  [ m =d m'; m' =d m'' ] ⇒ m =d m''
  {proof}

```

abbreviation $SRabber :: ('exp, 'id, 'val, 'com) \text{ TLSteps-curry}$
 $((1\langle\cdot,\cdot\rangle) \rightarrow \triangleleft\triangleright / (1\langle\cdot,\cdot\rangle) [0,0,0,0] 81)$

where

$$\langle c, m \rangle \rightarrow \triangleleft\alpha\triangleright \langle p, m' \rangle \equiv ((c, m), \alpha, (p, m')) \in SR$$

— function for obtaining the unique memory (state) after one step for a command and a memory (state)

definition $NextMem :: 'com \Rightarrow ('id, 'val) State \Rightarrow ('id, 'val) State$

($\llbracket c \rrbracket '(-)$)

where

$$\llbracket c \rrbracket (m) \equiv (\text{THE } m'. (\exists p \alpha. \langle c, m \rangle \rightarrow \triangleleft\alpha\triangleright \langle p, m' \rangle))$$

— function getting all escape hatches for some location

definition $htchLoc :: nat \Rightarrow ('d, 'exp) Hatches$

where

$$htchLoc \iota \equiv \{(d, e). (d, e, \iota) \in LH\}$$

— function for getting all escape hatches for some set of locations

definition $htchLocSet :: nat set \Rightarrow ('d, 'exp) Hatches$

where

$$htchLocSet PP \equiv \bigcup \{h. (\exists \iota \in PP. h = htchLoc \iota)\}$$

— predicate for (d,H)-equality

definition $dH-equal :: 'd \Rightarrow ('d, 'exp) Hatches$

$\Rightarrow ('id, 'val) State \Rightarrow ('id, 'val) State \Rightarrow bool$

where

$$dH-equal d H m m' \equiv (m =_d m' \wedge \\ (\forall (d', e) \in H. (d' \leq d \rightarrow (E e m = E e m'))))$$

abbreviation $dH-equal' :: ('id, 'val) State \Rightarrow 'd \Rightarrow ('d, 'exp) Hatches$

$\Rightarrow ('id, 'val) State \Rightarrow bool$

($(\cdot \sim_{-, -} \cdot)$)

where

$$m \sim_{d, H} m' \equiv dH-equal d H m m'$$

— predicate indicating that a command is not a d-declassification command

definition $NDC :: 'd \Rightarrow 'com \Rightarrow bool$

where

$$NDC d c \equiv (\forall m m'. m =_d m' \longrightarrow \llbracket c \rrbracket (m) =_d \llbracket c \rrbracket (m'))$$

— predicate indicating an 'immediate d-declassification command' for a set of escape hatches

definition $IDC :: 'd \Rightarrow 'com \Rightarrow ('d, 'exp) Hatches \Rightarrow bool$

where

$$IDC d c H \equiv (\exists m m'. m =_d m' \wedge \\ (\neg \llbracket c \rrbracket (m) =_d \llbracket c \rrbracket (m')))$$

$\wedge (\forall m m'. m \sim_{d,H} m' \longrightarrow \llbracket c \rrbracket(m) =_d \llbracket c \rrbracket(m'))$

definition *stepResultsinR* :: 'com ProgramState \Rightarrow 'com ProgramState
 \Rightarrow 'com Bisimulation-type \Rightarrow bool
where
stepResultsinR p p' R \equiv $(p = \text{None} \wedge p' = \text{None}) \vee$
 $(\exists c c'. (p = \text{Some } c \wedge p' = \text{Some } c' \wedge ([c],[c']) \in R))$

definition *dhequality-alternative* :: 'd \Rightarrow nat set \Rightarrow nat
 \Rightarrow ('id, 'val) State \Rightarrow ('id, 'val) State \Rightarrow bool
where
dhequality-alternative d PP i m m' \equiv $m \sim_{d,(htchLocSet PP)} m' \vee$
 $(\neg (htchLoc i) \subseteq (htchLocSet PP))$

definition *Strong-dlHPP-Bisimulation* :: 'd \Rightarrow nat set
 \Rightarrow 'com Bisimulation-type \Rightarrow bool
where
Strong-dlHPP-Bisimulation d PP R \equiv
 $(\text{sym } R) \wedge (\text{trans } R) \wedge$
 $(\forall (V, V') \in R. \text{length } V = \text{length } V') \wedge$
 $(\forall (V, V') \in R. \forall i < \text{length } V.$
 $((NDC d (V!i)) \vee$
 $(IDC d (V!i) (htchLoc (pp (V!i)))))) \wedge$
 $(\forall (V, V') \in R. \forall i < \text{length } V. \forall m1 m1' m2 \alpha p.$
 $(\langle V!i, m1 \rangle \rightarrow \triangleleft \alpha \triangleright \langle p, m2 \rangle \wedge m1 \sim_{d,(htchLocSet PP)} m1')$
 $\longrightarrow (\exists p' \alpha' m2'. (\langle V!i, m1 \rangle \rightarrow \triangleleft \alpha' \triangleright \langle p', m2 \rangle \wedge$
 $(\text{stepResultsinR } p p' R) \wedge (\alpha, \alpha') \in R \wedge$
 $(\text{dhequality-alternative } d PP (pp (V!i)) m2 m2'))))$

— predicate to define when a program is strongly secure

definition *WHATWHERE-Secure* :: 'com list \Rightarrow bool
where
WHATWHERE-Secure V \equiv $(\forall d PP.$
 $(\exists R. \text{Strong-dlHPP-Bisimulation } d PP R \wedge (V, V) \in R))$

— auxiliary lemma to obtain central strong (d,IH,PP)-Bisimulation property as Lemma in meta logic (allows instantiating all the variables manually if necessary)

lemma *strongdlHPPB-aux*:
 $\begin{aligned} &\wedge V V' m1 m1' m2 p i \alpha. \llbracket \text{Strong-dlHPP-Bisimulation } d PP R; \\ &i < \text{length } V; (V, V') \in R; \\ &\langle V!i, m1 \rangle \rightarrow \triangleleft \alpha \triangleright \langle p, m2 \rangle; m1 \sim_{d,(htchLocSet PP)} m1' \rrbracket \\ &\implies (\exists p' \alpha' m2'. \langle V!i, m1 \rangle \rightarrow \triangleleft \alpha' \triangleright \langle p', m2 \rangle \\ &\wedge \text{stepResultsinR } p p' R \wedge (\alpha, \alpha') \in R \wedge \\ &(dhequality-alternative d PP (pp (V!i)) m2 m2')) \end{aligned}$

lemma *strongdlHPPB-NDCIDCaux*:

$$\begin{aligned} & \wedge V V' i. \llbracket \text{Strong-dlHPP-Bisimulation } d \text{ PP } R; \\ & \quad (V, V') \in R; i < \text{length } V \rrbracket \\ & \quad \implies (\text{NDC } d (V!i) \vee \text{IDC } d (V!i) (\text{htchLoc } (\text{pp } (V!i)))) \\ & \langle \text{proof} \rangle \end{aligned}$$

lemma *WHATWHERE-empty*:
WHATWHERE-Secure []
 $\langle \text{proof} \rangle$

end

end

2.2 Proof technique for compositionality results

For proving compositionality results for WHAT&WHERE-security, we formalize the following “up-to technique” and prove it sound:

```
theory Up-To-Technique
imports WHATWHERE-Security
begin

context WHATWHERE
begin

abbreviation SdlHPPB where SdlHPPB  $\equiv$  Strong-dlHPP-Bisimulation
```

— define the ‘reflexive part’ of a relation (sets of elements which are related with themselves by the given relation)

definition Arefl :: ($'a \times 'a$) set $\Rightarrow 'a$ set

where

Arefl $R = \{e. (e,e) \in R\}$

```
lemma commonArefl-subset-commonDomain:
(Arefl  $R1 \cap$  Arefl  $R2) \subseteq (\text{Domain } R1 \cap \text{Domain } R2)$ 
 $\langle \text{proof} \rangle$ 
definition disj-dlHPP-Bisimulation-Up-To-R' :: 
'd  $\Rightarrow$  nat set  $\Rightarrow$  'com Bisimulation-type
 $\Rightarrow$  'com Bisimulation-type  $\Rightarrow$  bool
where
disj-dlHPP-Bisimulation-Up-To-R'  $d$  PP  $R'$   $R \equiv$ 
SdlHPPB  $d$  PP  $R'$   $\wedge$  (sym  $R$ )  $\wedge$  (trans  $R$ )
 $\wedge (\forall (V, V') \in R. \text{length } V = \text{length } V') \wedge$ 
 $(\forall (V, V') \in R. \forall i < \text{length } V.$ 
 $((\text{NDC } d (V!i)) \vee$ 
 $(\text{IDC } d (V!i) (\text{htchLoc } (\text{pp } (V!i)))))) \wedge$ 
 $(\forall (V, V') \in R. \forall i < \text{length } V. \forall m1 m1' m2 \alpha p.$ 
 $(\langle V!i, m1 \rangle \rightarrow \triangleleft \alpha \triangleright \langle p, m2 \rangle \wedge m1 \sim_{d, (\text{htchLocSet } PP)} m1')$ 
```

```

→ (exists p' alpha' m2'. ⟨V!i,m1⟩ →▫α▫ ⟨p',m2⟩ ∧
  (stepResultsinR p p' (R ∪ R')) ∧ (alpha, alpha') ∈ (R ∪ R') ∧
  (dhequality-alternative d PP (pp (V!i)) m2 m2)))

```

— lemma about the transitivity of the union of symmetric and transitive relations under certain circumstances

lemma *trans-RuR'*:

```

assumes transR: trans R
assumes symR: sym R
assumes transR': trans R'
assumes symR': sym R'
assumes eqlenR: ∀(V,V') ∈ R. length V = length V'
assumes eqlenR': ∀(V,V') ∈ R'. length V = length V'
assumes AreflAssump: (Arefl R ∩ Arefl R') ⊆ {[]}
shows trans (R ∪ R')
⟨proof⟩

```

lemma *Up-To-Technique*:

```

[ disj-dlHPP-Bisimulation-Up-To-R' d PP R' R;
  Arefl R ∩ Arefl R' ⊆ {[]} ]
  ==> SdlHPPB d PP (R ∪ R')
⟨proof⟩

```

lemma *Union-Strong-dlHPP-Bisim*:

```

[ SdlHPPB d PP R; SdlHPPB d PP R';
  Arefl R ∩ Arefl R' ⊆ {[]} ]
  ==> SdlHPPB d PP (R ∪ R')
⟨proof⟩

```

lemma *adding-emptypair-keeps-SdlHPPB*:

```

assumes SdlHPP: SdlHPPB d PP R
shows SdlHPPB d PP (R ∪ {[],[]})
⟨proof⟩

```

end

end

2.3 Proof of parallel compositionality

We prove that WHAT&WHERE-security is preserved under composition of WHAT&WHERE-secure threads.

```

theory Parallel-Composition
imports Up-To-Technique MWLs
begin

```

```

locale WHATWHERE-Secure-Programs =
   $L? : MWLs\text{-semantics } E \text{ } BMap$ 
  +  $WWs? : \text{WHATWHERE } MWLsSteps\text{-det } E \text{ } pp \text{ } DA \text{ } lH$ 
  for  $E :: ('exp, 'id, 'val)$  Evalfunction
  and  $BMap :: 'val \Rightarrow \text{bool}$ 
  and  $DA :: ('id, 'd::order)$  DomainAssignment
  and  $lH :: ('d, 'exp)$  lHatches
  begin

  lemma SdlHPPB-restricted-on-PP-is-SdlHPPB:
    assumes  $SdlHPPB : SdlHPPB d PP R'$ 
    assumes  $inR' : (V, V) \in R'$ 
    assumes  $Rdef : R = \{(V', V''). (V', V'') \in R' \wedge set(PPV V') \subseteq set(PPV V) \wedge set(PPV V'') \subseteq set(PPV V)\}$ 
    shows  $SdlHPPB d PP R$ 
   $\langle proof \rangle$ 

  theorem parallel-composition:
     $\llbracket \forall i < \text{length } V. \text{WHATWHERE-Secure } [V!i]; \text{unique-PPV } V \rrbracket$ 
     $\implies \text{WHATWHERE-Secure } V$ 
   $\langle proof \rangle$ 

  end
  end

```

3 Example language and compositionality proofs

3.1 Example language with dynamic thread creation

As in [2], we instantiate the language with a simple while language that supports dynamic thread creation via a spawn command (Multi-threaded While Language with spawn, MWLs). Note that the language is still parametric in the language used for Boolean and arithmetic expressions ('exp).

```

theory MWLs
imports Strong-Security.Types
begin

— type parameters not instantiated:
— 'exp: expressions (arithmetic, boolean...)
— 'val: numbers, boolean constants....
— 'id: identifier names
— SYNTAX

```

```

datatype ('exp, 'id) MWLsCom
= Skip nat (skip_ [50] 70)
| Assign 'id nat 'exp
  (-:=_ - [70,50,70] 70)
| Seq ('exp, 'id) MWLsCom
  ('exp, 'id) MWLsCom
  (-;- [61,60] 60)
| If-Else nat 'exp ('exp, 'id) MWLsCom
  ('exp, 'id) MWLsCom
  (if_ - then - else - fi [50,80,79,79] 70)
| While-Do nat 'exp ('exp, 'id) MWLsCom
  (while_ - do - od [50,80,79] 70)
| Spawn nat (('exp, 'id) MWLsCom) list
  (spawn_ - [50,70] 70)

```

— function for obtaining the program point of some MWLsloc command

primrec pp :: ('exp, 'id) MWLsCom \Rightarrow nat

where

```

pp (skipt) =  $\iota$  |
pp (x :=t e) =  $\iota$  |
pp (c1;c2) = pp c1 |
pp (ift b then c1 else c2 fi) =  $\iota$  |
pp (whilet b do c od) =  $\iota$  |
pp (spawnt V) =  $\iota$ 

```

— mutually recursive functions to collect program points of commands and thread pools

primrec PPc :: ('exp, 'id) MWLsCom \Rightarrow nat list

and PPV :: ('exp, 'id) MWLsCom list \Rightarrow nat list

where

```

PPc (skipt) = [t] |
PPc (x :=t e) = [t] |
PPc (c1;c2) = (PPc c1) @ (PPc c2) |
PPc (ift b then c1 else c2 fi) = [t] @ (PPc c1) @ (PPc c2) |
PPc (whilet b do c od) = [t] @ (PPc c) |
PPc (spawnt V) = [t] @ (PPV V) |

```

```

PPV [] = [] |
PPV (c# V) = (PPc c) @ (PPV V)

```

— predicate indicating that a command only contains unique program points

definition unique-PPc :: ('exp, 'id) MWLsCom \Rightarrow bool

where

unique-PPc c = distinct (PPc c)

— predicate indicating that a thread pool only contains unique program points

definition *unique-PPV* :: ('exp, 'id) MWLsCom list \Rightarrow bool

where

unique-PPV V = *distinct* (*PPV V*)

lemma *PPc-nonempt*: *PPc c* $\neq \emptyset$
{proof}

lemma *unique-c-uneq*: set (*PPc c*) \cap set (*PPc c'*) = {} \Rightarrow *c* \neq *c'*
{proof}

lemma *V-nonempt-PPV-nonempt*: *V* $\neq \emptyset$ \Rightarrow *PPV V* $\neq \emptyset$
{proof}

lemma *unique-V-uneq*:
 $\llbracket V \neq \emptyset; V' \neq \emptyset; \text{set}(\text{PPV } V) \cap \text{set}(\text{PPV } V') = \{\} \rrbracket \Rightarrow V \neq V'$
{proof}

lemma *PPc-in-PPV*: *c* \in set *V* \Rightarrow set (*PPc c*) \subseteq set (*PPV V*)
{proof}

lemma *listindices-aux*: *i* $<$ length *V* \Rightarrow (*V!i*) \in set *V*
{proof}

lemma *PPc-in-PPV-version*:
 $i < \text{length } V \Rightarrow \text{set}(\text{PPc } (\text{V!i})) \subseteq \text{set}(\text{PPV } V)$
{proof}

lemma *uniPPV-uniPPc*: *unique-PPV V* \Rightarrow ($\forall i < \text{length } V. \text{unique-PPc } (\text{V!i})$)
{proof}

locale *MWLs-semantics* =
fixes *E* :: ('exp, 'id, 'val) Evalfunction
and *BMap* :: 'val \Rightarrow bool
begin

— steps semantics, set of deterministic steps from commands to program states

inductive-set

MWLsSteps-det ::
('exp, 'id, 'val, ('exp, 'id) MWLsCom) TLSteps

and *MWLslocSteps-det'* ::
('exp, 'id, 'val, ('exp, 'id) MWLsCom) TLSteps-curry
 $((1\langle\cdot,\cdot\rangle) \rightarrow \triangleleft\triangleright / (1\langle\cdot,\cdot\rangle) [0,0,0,0,0] 81)$

where

$\langle c1, m1 \rangle \rightarrow \triangleleft\alpha\triangleright \langle c2, m2 \rangle \equiv ((c1, m1), \alpha, (c2, m2)) \in \text{MWLsSteps-det} \mid$
skip: $\langle \text{skip}, m \rangle \rightarrow \triangleleft\emptyset\triangleright \langle \text{None}, m \rangle \mid$
assign: $(E e m) = v \Rightarrow$
 $\langle x :=_t e, m \rangle \rightarrow \triangleleft\emptyset\triangleright \langle \text{None}, m(x := v) \rangle \mid$
seq1: $\langle c1, m \rangle \rightarrow \triangleleft\alpha\triangleright \langle \text{None}, m' \rangle \Rightarrow$

```

 $\langle c1;c2,m \rangle \rightarrow \triangleleft \alpha \triangleright \langle Some\ c2,m' \rangle |$ 
seq2:  $\langle c1,m \rangle \rightarrow \triangleleft \alpha \triangleright \langle Some\ c1',m' \rangle \implies$ 
 $\langle c1;c2,m \rangle \rightarrow \triangleleft \alpha \triangleright \langle Some\ (c1';c2),m' \rangle |$ 
iftrue:  $BMap\ (E\ b\ m) = True \implies$ 
 $\langle if_{\iota}\ b\ then\ c1\ else\ c2\ fi,m \rangle \rightarrow \triangleleft \square \triangleright \langle Some\ c1,m \rangle |$ 
iffalse:  $BMap\ (E\ b\ m) = False \implies$ 
 $\langle if_{\iota}\ b\ then\ c1\ else\ c2\ fi,m \rangle \rightarrow \triangleleft \square \triangleright \langle Some\ c2,m \rangle |$ 
whiletrue:  $BMap\ (E\ b\ m) = True \implies$ 
 $\langle while_{\iota}\ b\ do\ c\ od,m \rangle \rightarrow \triangleleft \square \triangleright \langle Some\ (c;(while_{\iota}\ b\ do\ c\ od)),m \rangle |$ 
whilefalse:  $BMap\ (E\ b\ m) = False \implies$ 
 $\langle while_{\iota}\ b\ do\ c\ od,m \rangle \rightarrow \triangleleft \square \triangleright \langle None,m \rangle |$ 
spawn:  $\langle spawn_{\iota}\ V,m \rangle \rightarrow \triangleleft V \triangleright \langle None,m \rangle$ 

```

inductive-cases MWLsSteps-det-cases:

```

 $\langle skip_{\iota},m \rangle \rightarrow \triangleleft \alpha \triangleright \langle p,m' \rangle$ 
 $\langle x :=_{\iota} e,m \rangle \rightarrow \triangleleft \alpha \triangleright \langle p,m' \rangle$ 
 $\langle c1;c2,m \rangle \rightarrow \triangleleft \alpha \triangleright \langle p,m' \rangle$ 
 $\langle if_{\iota}\ b\ then\ c1\ else\ c2\ fi,m \rangle \rightarrow \triangleleft \alpha \triangleright \langle p,m' \rangle$ 
 $\langle while_{\iota}\ b\ do\ c\ od,m \rangle \rightarrow \triangleleft \alpha \triangleright \langle p,m' \rangle$ 
 $\langle spawn_{\iota}\ V,m \rangle \rightarrow \triangleleft \alpha \triangleright \langle p,m' \rangle$ 

```

— non-deterministic, probabilistic system step (added for intuition, not used in the proofs)

inductive-set

MWLsSteps-ndet ::

```

('exp, 'id, 'val, ('exp, 'id) MWLsCom) TPSteps
and MWLsSteps-ndet' :: 
    ('exp, 'id, 'val, ('exp, 'id) MWLsCom) TPSteps-curry
 $((1\langle\ ,/\,\rangle) \Rightarrow / (1\langle\ ,/\,\rangle) [0,0,0,0] 81)$ 

```

where

```

 $\langle V,m \rangle \Rightarrow \langle V',m' \rangle \equiv ((V,m),(V',m')) \in MWLsSteps-ndet |$ 
stepthreadi1:  $\langle ci,m \rangle \rightarrow \triangleleft \alpha \triangleright \langle None,m' \rangle \implies$ 
 $\langle cf @ [ci] @ ca,m \rangle \Rightarrow \langle cf @ \alpha @ ca,m' \rangle |$ 
stepthreadi2:  $\langle ci,m \rangle \rightarrow \triangleleft \alpha \triangleright \langle Some\ c',m' \rangle \implies$ 
 $\langle cf @ [ci] @ ca,m \rangle \Rightarrow \langle cf @ [c] @ \alpha @ ca,m \rangle$ 

```

— lemma about existence and uniqueness of next memory of a step

lemma nextmem-exists-and-unique:

```

 $\exists m'\ p\ \alpha.\ \langle c,m \rangle \rightarrow \triangleleft \alpha \triangleright \langle p,m' \rangle$ 
 $\wedge (\forall m''. (\exists p\ \alpha.\ \langle c,m \rangle \rightarrow \triangleleft \alpha \triangleright \langle p,m'' \rangle) \longrightarrow m'' = m')$ 
 $\langle proof \rangle$ 

```

lemma PPsc-of-step:

```

 $\llbracket \langle c,m \rangle \rightarrow \triangleleft \alpha \triangleright \langle p,m' \rangle; \exists c'. p = Some\ c' \rrbracket$ 
 $\implies set\ (PPc\ (the\ p)) \subseteq set\ (PPc\ c)$ 
 $\langle proof \rangle$ 

```

lemma PPs α -of-step:

```

 $\langle c, m \rangle \rightarrow \lhd \alpha \rhd \langle p, m' \rangle$ 
 $\implies \text{set}(\text{PPV } \alpha) \subseteq \text{set}(\text{PPc } c)$ 
 $\langle \text{proof} \rangle$ 

```

end

end

3.2 Proofs of atomic compositionality results

We prove for each atomic command of our example programming language (i.e. a command that is not composed out of other commands) that it is strongly secure if the expressions involved are indistinguishable for an observer on security level d .

```

theory WHATWHERE-Secure-Skip-Assign
imports Parallel-Composition
begin

```

```

context WHATWHERE-Secure-Programs
begin

```

```

abbreviation NextMem'
( $\llbracket \cdot \rrbracket$ )'
where
 $\llbracket c \rrbracket(m) \equiv \text{NextMem } c \ m$ 

```

— define when two expressions are indistinguishable with respect to a domain d

```

definition d-indistinguishable :: 'd::order  $\Rightarrow$  'exp  $\Rightarrow$  'exp  $\Rightarrow$  bool
where
d-indistinguishable  $d \ e1 \ e2 \equiv \forall m \ m'. ((m =_d m') \rightarrow ((E \ e1 \ m) = (E \ e2 \ m')))$ 

```

```

abbreviation d-indistinguishable' :: 'exp  $\Rightarrow$  'd::order  $\Rightarrow$  'exp  $\Rightarrow$  bool
(  $(\cdot \equiv_{\cdot} \cdot)$  )

```

```

where
 $e1 \equiv_d e2 \equiv \text{d-indistinguishable } d \ e1 \ e2$ 

```

— symmetry of d-indistinguishable

lemma d-indistinguishable-sym:

```

 $e \equiv_d e' \implies e' \equiv_d e$ 
 $\langle \text{proof} \rangle$ 

```

lemma d-indistinguishable-trans:

```

 $\llbracket e \equiv_d e'; e' \equiv_d e'' \rrbracket \implies e \equiv_d e''$ 
 $\langle \text{proof} \rangle$ 

```

```

definition dH-indistinguishable :: 'd  $\Rightarrow$  ('d, 'exp) Hatches
 $\Rightarrow$  'exp  $\Rightarrow$  'exp  $\Rightarrow$  bool

```

```

where
 $dH\text{-indistinguishable } d H e1 e2 \equiv (\forall m m'. m \sim_{d,H} m' \rightarrow (E e1 m = E e2 m'))$ 

abbreviation  $dH\text{-indistinguishable}' :: 'exp \Rightarrow 'd \Rightarrow ('d, 'exp) Hatches \Rightarrow 'exp \Rightarrow bool$   

 $((- \equiv_{-, -} -))$ 
where
 $e1 \equiv_{d,H} e2 \equiv dH\text{-indistinguishable } d H e1 e2$ 

lemma  $\text{emp}_H\text{-implies-}dH\text{-indistinguishable-eq-dindistinguishable}$ :  

 $(e \equiv_{d,\{\}} e') = (e \equiv_d e')$   

 $\langle proof \rangle$ 

theorem  $\text{WHATWHERE-Secure-Skip}$ :  

 $\text{WHATWHERE-Secure } [\text{skip}_t]$   

 $\langle proof \rangle$ 
lemma  $\text{semAssignSC-aux}$ :  

assumes  $\text{dhind}: e \equiv_{DA} x,(\text{htchLoc } t) e$   

shows  $\text{NDC } d (x :=_t e) \vee \text{IDC } d (x :=_t e) (\text{htchLoc } (\text{pp } (x :=_t e)))$   

 $\langle proof \rangle$ 

theorem  $\text{WHATWHERE-Secure-Assign}$ :  

assumes  $\text{dhind}: e \equiv_{DA} x,(\text{htchLoc } t) e$   

assumes  $\text{dheq-imp}: \forall m m' d t'. (m \sim_{d,(\text{htchLoc } t')} m' \wedge$   

 $\llbracket x :=_t e \rrbracket(m) =_d \llbracket x :=_t e \rrbracket(m')) \rightarrow \llbracket x :=_t e \rrbracket(m) \sim_{d,(\text{htchLoc } t')} \llbracket x :=_t e \rrbracket(m')$   

shows  $\text{WHATWHERE-Secure } [x :=_t e]$   

 $\langle proof \rangle$ 

end  

end

```

3.3 Proofs of non-atomic compositionality results

We prove compositionality results for each non-atomic command of our example programming language (i.e. a command that is composed out of other commands): If the components are strongly secure and the expressions involved indistinguishable for an observer on security level d , then the composed command is also strongly secure.

```

theory Language-Composition
imports  $\text{WHATWHERE-Secure-Skip-Assign}$ 
begin

```

```

context WHATWHERE-Secure-Programs
begin

theorem Compositionality-Seq:
  assumes WWs-part1: WHATWHERE-Secure [c1]
  assumes WWs-part2: WHATWHERE-Secure [c2]
  assumes uniPPc1c2: unique-PPc (c1;c2)
  shows WHATWHERE-Secure [c1;c2]
  ⟨proof⟩

theorem Compositionality-Spawn:
  assumes WWs-threads: WHATWHERE-Secure V
  assumes uniPPspawn: unique-PPc (spawnt V)
  shows WHATWHERE-Secure [spawnt V]
  ⟨proof⟩

theorem Compositionality-If:
  assumes dind: ∀ d. b ≡d b
  assumes WWs-branch1: WHATWHERE-Secure [c1]
  assumes WWs-branch2: WHATWHERE-Secure [c2]
  assumes uniPPif: unique-PPc (ift b then c1 else c2 fi)
  shows WHATWHERE-Secure [ift b then c1 else c2 fi]
  ⟨proof⟩

theorem Compositionality-While:
  assumes dind: ∀ d. b ≡d b
  assumes WWs-body: WHATWHERE-Secure [c]
  assumes uniPPwhile: unique-PPc (whilet b do c od)
  shows WHATWHERE-Secure [whilet b do c od]
  ⟨proof⟩

end

end

```

4 Security type system

4.1 Abstract security type system with soundness proof

We formalize an abstract version of the type system in [2] using locales [1]. Our formalization of the type system is abstract in the sense that the rules specify abstract semantic side conditions on the expressions within a command that satisfy for proving the soundness of the rules. That is, it can be instantiated with different syntactic approximations for these semantic

side conditions in order to achieve a type system for a concrete language for Boolean and arithmetic expressions. Obtaining a soundness proof for such a concrete type system then boils down to proving that the concrete type system interprets the abstract type system.

We prove the soundness of the abstract type system by simply applying the compositionality results proven before.

```

theory Type-System
imports Language-Composition
begin

locale Type-System =
  WWP?: WHATWHERE-Secure-Programs E BMap DA lH
  for E :: ('exp, 'id, 'val) Evalfunction
  and BMap :: 'val ⇒ bool
  and DA :: ('id, 'd::order) DomainAssignment
  and lH :: ('d, 'exp) lHatches
+
fixes
  AssignSideCondition :: 'id ⇒ 'exp ⇒ nat ⇒ bool
  and WhileSideCondition :: 'exp ⇒ bool
  and IfSideCondition :: 
    'exp ⇒ ('exp,'id) MWLsCom ⇒ ('exp,'id) MWLsCom ⇒ bool
  assumes semAssignSC: AssignSideCondition x e i ⇒
    e ≡DA x, (htchLoc i) e ∧ (∀ m m' d i'. (m ~d, (htchLoc i') m' ∧
    [x :=i e](m) =d [x :=i e](m')) → [x :=i e](m) ~d, (htchLoc i') [x :=i e](m'))
  and semWhileSC: WhileSideCondition e ⇒ ∀ d. e ≡d e
  and semIfSC: IfSideCondition e c1 c2 ⇒ ∀ d. e ≡d e
begin

— Security typing rules for the language commands
inductive
  ComSecTyping :: ('exp, 'id) MWLsCom ⇒ bool
  (⊢C -)
  and ComSecTypingL :: ('exp, 'id) MWLsCom list ⇒ bool
  (⊢V -)
  where
    Skip: ⊢C skipi |
    Assign: [ AssignSideCondition x e i ] ⇒ ⊢C x :=i e |
    Spawn: [ ⊢V V ] ⇒ ⊢C spawni V |
    Seq: [ ⊢C c1; ⊢C c2 ] ⇒ ⊢C c1;c2 |
    While: [ ⊢C c; WhileSideCondition b ] ⇒ ⊢C whilei b do c od |
    If: [ ⊢C c1; ⊢C c2; IfSideCondition b c1 c2 ] ⇒ ⊢C ifi b then c1 else c2 fi |
    Parallel: [ ∀ i < length V. ⊢C V!i ] ⇒ ⊢V V
  
```

```

inductive-cases parallel-cases:
 $\vdash_{\mathcal{V}} V$ 

definition auxiliary-predicate
where
auxiliary-predicate  $V \equiv \text{unique-PPV } V \longrightarrow \text{WHATWHERE-Secure } V$ 

```

— soundness proof of abstract type system

theorem ComSecTyping-single-is-sound:

$$\begin{aligned} & [\vdash_{\mathcal{C}} c; \text{unique-PPc } c] \\ & \implies \text{WHATWHERE-Secure } [c] \end{aligned}$$

$\langle \text{proof} \rangle$

theorem ComSecTyping-list-is-sound:
 $[\vdash_{\mathcal{V}} V; \text{unique-PPV } V] \implies \text{WHATWHERE-Secure } V$
 $\langle \text{proof} \rangle$

end

end

4.2 Example language for Boolean and arithmetic expressions

As an example, we provide a simple example language for instantiating the parameter ' \exp ' for the language for Boolean and arithmetic expressions (from the entry Strong-Security).

```

theory Expr
imports Types
begin

— type parameters:
— 'val': numbers, boolean constants....
— 'id': identifier names

type-synonym ('val) operation = 'val list  $\Rightarrow$  'val

datatype (dead 'id, dead 'val) Expr =
Const 'val |
Var 'id |
Op 'val operation (('id, 'val) Expr) list

```

— defining a simple recursive evaluation function on this datatype
primrec ExprEval :: (('id, 'val) Expr, 'id, 'val) Evalfunction
and ExprEvalL :: (('id, 'val) Expr) list \Rightarrow ('id, 'val) State \Rightarrow 'val list

```

where
ExprEval (Const v) m = v |
ExprEval (Var x) m = (m x) |
ExprEval (Op f arglist) m = (f (ExprEvalL arglist m)) |

ExprEvalL [] m = [] |
ExprEvalL (e#V) m = (ExprEval e m) #(ExprEvalL V m)

end

```

4.3 Example interpretation of abstract security type system

Using the example instantiation of the language for Boolean and arithmetic expressions, we give an example instantiation of our abstract security type system, instantiating the parameter for domains '*d*' with a two-level security lattice (from the entry Strong-Security).

```

theory Domain-example
imports Expr
begin

— When interpreting, we have to instantiate the type for domains. As an example,
we take a type containing 'low' and 'high' as domains.

datatype Dom = low | high

instantiation Dom :: order
begin

definition
less-eq-Dom-def: d1 ≤ d2 = (if d1 = d2 then True
else (if d1 = low then True else False))

definition
less-Dom-def: d1 < d2 = (if d1 = d2 then False
else (if d1 = low then True else False))

instance ⟨proof⟩

end

end

```

```

theory Type-System-example
imports Type-System Strong-Security Expr Strong-Security.Domain-example
begin

```

— When interpreting, we have to instantiate the type for domains. As an example,

we take a type containing 'low' and 'high' as domains.

```
consts DA :: ('id,Dom) DomainAssignment
consts BMap :: 'val ⇒ bool
consts lH :: (Dom,('id,'val) Expr) lHatches
```

— redefine all the abbreviations necessary for auxiliary lemmas with the correct parameter instantiation

```
abbreviation MWLsStepsdet' ::  

  (('id,'val) Expr, 'id, 'val, (('id,'val) Expr,'id) MWLsCom) TLSteps-curry  

  (((1⟨-,/-⟩) → ◁-▷ / (1⟨-,/-⟩) [0,0,0,0,0] 81))  

where  

  ⟨c1,m1⟩ → ◁α▷ ⟨c2,m2⟩ ≡  

  ((c1,m1),α,(c2,m2)) ∈ MWLs-semantics.MWLsSteps-det ExprEval BMap
```

```
abbreviation d-equal' :: ('id, 'val) State  

  ⇒ Dom ⇒ ('id, 'val) State ⇒ bool  

  ( ( - =_ - ) )  

where  

  m =d m' ≡ WHATWHERE.d-equal DA d m m'
```

```
abbreviation dH-equal' :: ('id, 'val) State ⇒ Dom  

  ⇒ (Dom,('id,'val) Expr) Hatches  

  ⇒ ('id, 'val) State ⇒ bool  

  ( ( - ~_,- - ) )  

where  

  m ~d,H m' ≡ WHATWHERE.dH-equal ExprEval DA d H m m'
```

```
abbreviation NextMem' :: (('id,'val) Expr, 'id) MWLsCom  

  ⇒ ('id,'val) State ⇒ ('id,'val) State  

  ([[-](-))  

where  

  [[c]](m)  

  ≡ WHATWHERE.NextMem (MWLs-semantics.MWLsSteps-det ExprEval BMap)  

  c m
```

```
abbreviation dH-indistinguishable' :: ('id,'val) Expr ⇒ Dom  

  ⇒ (Dom,('id,'val) Expr) Hatches ⇒ ('id,'val) Expr ⇒ bool  

  ( ( - ≡_,- - ) )  

where  

  e1 ≡d,H e2  

  ≡ WHATWHERE-Secure-Programs.dH-indistinguishable ExprEval DA d H e1 e2
```

```
abbreviation htchLoc :: nat ⇒ (Dom, ('id,'val) Expr) Hatches  

where  

  htchLoc i ≡ WHATWHERE.htchLoc lH i
```

— Security typing rules for expressions

inductive

```
ExprSecTyping :: (Dom, ('id,'val) Expr) Hatches
  ⇒ ('id, 'val) Expr ⇒ Dom ⇒ bool
  (- ⊢E - : -)
for H :: (Dom, ('id, 'val) Expr) Hatches
where
  Consts: H ⊢E (Const v) : d |
  Vars: DA x = d ⇒ H ⊢E (Var x) : d |
  Hatch: (d,e) ∈ H ⇒ H ⊢E e : d |
  Ops: [ ∀ i < length arglist. H ⊢E (arglist!i) : (dl!i) ∧ (dl!i) ≤ d ]
    ⇒ H ⊢E (Op f arglist) : d
```

— function substituting a certain expression with another expression in expressions

```
primrec Subst :: ('id, 'val) Expr ⇒ ('id, 'val) Expr
  ⇒ ('id, 'val) Expr ⇒ ('id, 'val) Expr
  (-<-＼->)
and SubstL :: ('id, 'val) Expr list ⇒ ('id, 'val) Expr
  ⇒ ('id, 'val) Expr ⇒ ('id, 'val) Expr list
where
  (Const v)<e1＼e2> = (if e1=(Const v) then e2 else (Const v)) |
  (Var x)<e1＼e2> = (if e1=(Var x) then e2 else (Var x)) |
  (Op f arglist)<e1＼e2> = (if e1=(Op f arglist) then e2 else
  (Op f (SubstL arglist e1 e2))) |
```

$$\text{SubstL } [] \text{ } e1 \text{ } e2 = [] |$$

$$\text{SubstL } (e \# V) \text{ } e1 \text{ } e2 = (e < e1 \setminus e2 >) \# (\text{SubstL } V \text{ } e1 \text{ } e2)$$

definition SubstClosure :: 'id ⇒ ('id, 'val) Expr ⇒ bool

where

$\text{SubstClosure } x \text{ } e \equiv \forall (d', e', \iota') \in LH. (d', e' < (Var x) \setminus e, \iota') \in LH$

definition synAssignSC :: 'id ⇒ ('id, 'val) Expr ⇒ nat ⇒ bool

where

$\text{synAssignSC } x \text{ } e \iota \equiv \exists d. ((htchLoc \iota) \vdash_{\mathcal{E}} e : d \wedge d \leq DA x)$
 $\wedge (\text{SubstClosure } x \text{ } e)$

definition synWhileSC :: ('id, 'val) Expr ⇒ bool

where

$\text{synWhileSC } e \equiv (\exists d. (\{\} \vdash_{\mathcal{E}} e : d) \wedge (\forall d'. d \leq d'))$

definition synIfSC :: ('id, 'val) Expr

```
  ⇒ (('id, 'val) Expr, 'id) MWLsCom
  ⇒ (('id, 'val) Expr, 'id) MWLsCom ⇒ bool
```

where

$\text{synIfSC } e \text{ } c1 \text{ } c2 \equiv \exists d. (\{\} \vdash_{\mathcal{E}} e : d \wedge (\forall d'. d \leq d'))$

— auxiliary lemma for locale interpretation (theorem 7 in original paper)

lemma ExprTypable-with-smallerd-implies-dH-indistinguishable:

$$[\![H \vdash_{\mathcal{E}} e : d'; d' \leq d]\!] \implies e \equiv_{d,H} e$$

(proof)

lemma substexp-substmem:

$$\begin{aligned} \textit{ExprEval } e' < \textit{Var } x \setminus e > m &= \textit{ExprEval } e' (m(x := \textit{ExprEval } e m)) \\ &\wedge \textit{ExprEvalL } (\textit{SubstL } \textit{elist } (\textit{Var } x) e) m \\ &= \textit{ExprEvalL } \textit{elist } (m(x := \textit{ExprEval } e m)) \end{aligned}$$

(proof)

lemma SubstClosure-implications:

$$\begin{aligned} [\![\textit{SubstClosure } x e; m \sim_{d,(\textit{htchLoc } \iota')} m']\!] ; \\ [\![x :=_{\iota} e]\!](m) =_d [\![x :=_{\iota} e]\!](m') \\ \implies [\![x :=_{\iota} e]\!](m) \sim_{d,(\textit{htchLoc } \iota')} [\![x :=_{\iota} e]\!](m') \end{aligned}$$

(proof)

interpretation Type-System-example: Type-System ExprEval BMap DA lH
synAssignSC synWhileSC synIfSC

(proof)

end

References

- [1] C. Ballarin. Locales and Locale Expressions in Isabelle/Isar. In S. Berardi, M. Coppo, and F. Damiani, editors, *TYPES*, volume 3085 of *Lecture Notes in Computer Science*, pages 34–50. Springer, 2003.
- [2] A. Lux, H. Mantel, and M. Perner. Scheduler-independent declassification. In *Proceedings of the 11th International Conference on Mathematics of Program Construction (MPC)*, LNCS 7342, pages 25–47. Springer, 2012.